# Grafy funkcí

Grafy funkcí se kreslí příkazem Plot, který má hodně voleb. Ukážeme z nich jen ty nejdůležitější. Nejdříve volba barvy, šířky a způsobu vykreslení grafu (kromě ukázaného Dashed lze volit i Dotted nebo DotDashed). Šířku grafu lze ovlivnit číslem n v Thickness[n] (jiná volba je Thin).

```
f[x_] := x^3 - 4 * x + 1
```

```
Plot[f[x], {x, -1, 3}]
```




```
Plot[f[x], {x, -1, 3}, PlotStyle → {Dashed}]
```



```
Plot[f[x], {x, -1, 3}, PlotStyle → {Red, Dashed}]
```




```
Plot[f[x], {x, -1, 3}, PlotStyle → {Green, Thickness[0.03]}]
```




Je-li funkce nespojitá, budou některé skoky spojeny v závislosti na hodnotě v bodě, ve kterém skok nastává (není to však zcela důsledné).

```
Plot[Tan[x], {x, -5, 5}]
```



```
Plot[Floor[x], {x, 0, 5}]
```




```
Plot[Abs[x - 1] / (x - 1), {x, -2, 2}]
```




```
Plot[Sign[x - 1], {x, -2, 2}]
```




Nechceme-li některé svislé spojnice v grafu mít, lze je vyloučit pomocí Exclusions.

```
Plot[Tan[x], {x, -5, 5}, Exclusions → {-Pi / 2, Pi / 2}]
```




```
Plot[Sign[x - 1], {x, -2, 2}, Exclusions → {1}]
```




Grafy funkcí definovaných po částech mají své zvláštnosti. Většinou byly znázorněny v předchozím souboru. Grady nezaznamenají hodnotu změněnou nebo dodefinovanou v jednom bodě.

## Volby pro grafy

Pomocí PlotRange lze měnit obor zobrazovaných hodnot, PlotPoints mění počet bodů, ve kterých se hodnoty funkce počítají (daná hodnota je 50, je vhodné použít i volbu MaxRecursion, která udává počet rekurzí při počítání hodnot bodů), Mesh na grafu udělá puntíky nebo čárky aj. podle volby MeshStyle, Filling vyplní jisté plochy podle zadání, Epilog přidá po nakreslení grafu další jednoduché objekty, AspectRatio určuje zachování stejného měřítka na obou souřadnicových osách.

```
Plot[Tan[x], {x, -Pi / 2, Pi / 2}, PlotRange → {-4, 20}]
```




```
Plot[Sin[x], {x, 0, 7}, Mesh → 20,
 MeshStyle → PointSize[Medium], PlotPoints → 10,
 MaxRecursion → 0]
```




Pro Filling jsou různé možnosti, např. vyplnění barvou k ose, nad graf, pod graf, mezi grafy.

```
Plot[Sin[x], {x, 0, 10}, Filling → Axis]
```




Použití AspectRatio je vidět z porovnání předchozího a následujícího grafu funkce sinus.

```
Plot[Sin[x], {x, 0, 10}, AspectRatio → Automatic]
```




Pomocí volby Epilog můžeme přidat označení, které koncové body grafu patří a které nikoli.

```
Clear[f0]; f0[x_] := Piecewise[{{-1, x < 1}, {1, x > 1}}];
Plot[f0[x], {x, -2, 2},
 Epilog → {{PointSize[0.03], Point[{1, 0}]},
   {EdgeForm[Black], White, Disk[{1, -1}, 0.06],
    Disk[{1, 1}, 0.06]}}]
```




## Grafy více funkcí

```
Plot[{Sin[x], Cos[x], x / 5}, {x, 0, 2 * Pi}]
```



```
Plot[{Sin[x], Cos[x], x / 5}, {x, 0, 2 * Pi},
 PlotStyle → {Dotted, Thickness[0.02], Green}]
```




## Grafy jinak zadaných funkcí

Parametricky zadané křivky se kreslí pomocí ParametricPlot. Volby jsou podobné jako u Plot.

```
ParametricPlot[{2 * Cos[t] - Cos[2 * t], 2 * Sin[t] - Sin[2 * t]},
 {t, 0, 2 * Pi}]
```



```
ParametricPlot[{{2 * Cos[t] - Cos[2 * t], 2 * Sin[t] - Sin[2 * t]},
  {2 * Cos[t], 2 * Sin[t]}}, {t, 0, 2 * Pi}]
```




Dva grafy parametricky zadaných křivek s různými definičními obory parametru mají problémy s vykreslením grafů v různých barvách.

```
ParametricPlot[{{2 * Cos[t] - Cos[2 * t], 2 * Sin[t] - Sin[2 * t]},
  {u, -u + 1}}, {t, 0, 2 * Pi}, {u, -2, 1}]
```



Polárně zadané křivky se kreslí pomocí příkazu PolarPlot. Další volby jsou podobné jako výše.

```
PolarPlot[-4 * (Cos[t / 2])^2, {t, -Pi, Pi}]
```




Implicitně zadané křivky se dříve kreslili pomnocí příkazu ImplicitPlot (předem je však nutné nahrát balíček ImplicitPlot).Nyní se používá příkaz ContourPlot. V obou případech se musí použít ==, protože program rozhoduje, zda podmínka rovnosti platí. Pro ContourPlot lze použít mnoho voleb z příkazu Plot i některé specifické. Na rozdíl od ContourPlot kreslí ImplicitPlot souřadnicové osy, jak jsme zvyklí.

```
<< Graphics`ImplicitPlot`
```

— *General::obspkg :*

*Graphics`ImplicitPlot` is now obsolete. The legacy version being loaded may conflict with current Mathematica functionality. See the Compatibility Guide for updating information. ≫*

```
Clear[x]; Clear[y];
ImplicitPlot[x^2 / 4 + y^2 == 1, {x, -2, 2}, {y, -1, 1},
 AspectRatio → Automatic]
```




```
ContourPlot[x^2 / 4 + y^2 == 1, {x, -2, 2}, {y, -1, 1},
 AspectRatio → Automatic]
```




## Posunutí a násobení funkce

Následující animace hezky ukáže, jak se mění graf funkce při posunutí nebo vynásobení nezávislé proměnné, další pak totéž pro závisle proměnnou.

```
Manipulate[Plot[Sin[a x + b], {x, 0, 2 * Pi},
  ImageSize → {600, 400}], {a, 1, 5}, {b, 0, 5}]
```




```
Manipulate[Plot[a * Sin[x] + b, {x, 0, 6}, PlotRange → {-4, 4},
  AspectRatio → Automatic, ImageSize → {600, 400}],
 {a, 1, 4}, {b, 0, 2}]
```